


1. Open OpenSCAD - This is a design program which does not require drawing skills, it will require a bit of thinking in 3D though!

2. The OpenSCAD window has three areas:


Left, the code, a set of descriptions of what we want it to draw;

Top right, a 3D view of the model;

Bottom right, the status of the model creation - look here for errors!



3) Setup: To follow along with these instructions, visit the "View" menu and make sure the "Show Axes" option is ticked. To make sure the model picture always shows your current code, look in the "Design" menu and make sure the "Automatic Reload and Compile" option is ticked.




4) Get started! Lets draw something - to make a cube we type "cube" followed by parentheses to indicate parameters. Inside the parentheses we tell it the size (length of the edges), like this:

```
cube(size = 5);
```

The end of a statement has a semi-colon ";" to show where it finishes.

To show what this does, use CTRL+s (or File, Save) and save the file under your name, make sure it has a ".scad" file ending.

If OpenSCAD only displays a square, click your left mouse button and drag the image a bit the left so that you see the cube.




To zoom in use the mouse wheel.

To move the image around, click the right mouse button and drag.

5) We can also draw cylinders, they need a height and a radius:


```
cylinder(h = 10, r = 4);
```

Save (CTRL-S) again to see the cylinder.




5b) The cylinder doesn't look very round, to fix it we can set the "\$fn" value in the code:

```
$fn = 50;  
cylinder(h = 10, r = 4);
```




6) Turn the cylinder into a cone by giving a radius for each end:

```
$fn = 50;  
cylinder(h = 10, r1 = 4,  
r2 = 0);
```



7) We could put a head on that and call it a board game piece. How about a sphere:

```
$fn = 50;  
cylinder(h = 10, r1 = 4,  
r2 = 0);  
  
sphere(r=2.5);
```




7b) Oops, that should go on top of the cone, to move it upwards we use "translate", we need to tell it the movement in each of the three possible directions: X, Y and Z. To just move up, use 0 for X and Y.

To move the sphere, put the translate before the sphere command:

```
$fn = 50;
cylinder(h = 10, r1 = 4, r2 = 0);

translate([0, 0, 9])
sphere(r=2.5);
```




8) We can also add some arms, using another small cylinder along the edge of the cone-body, first add the cylinder:

```
$fn = 50;
// body
cylinder(h = 10, r1 = 4, r2 = 0);

// head
translate([0, 0, 9])
sphere(r=2.5);

// arms
cylinder(h = 4, r = 1);
```




I've also added some comments using "//" to help keep track of the pieces.

8b) That cylinder didn't appear? It's been drawn inside our cone body, use a "*" to hide the other parts for the moment:

```
$fn = 50;
// body
*cylinder(h = 10, r1 = 4,
r2 = 0);

// head
*translate([0, 0, 9])
  sphere(r=2.5);

// arms
cylinder(h = 4, r = 1);
```




9) Let's add an easy hand as well, a small sphere will do it, it'll appear at the end of the arm. Spheres have a radius (r) just like a cylinder:

```
$fn = 50;
// body
*cylinder(h = 10, r1 = 4,
r2 = 0);

// head
*translate([0, 0, 9])
  sphere(r=2.5);

// arms
cylinder(h = 4, r = 1);
sphere(r=1);
```




10) The arm needs to move to lie alongside the edges of the cone, so we need to rotate it, rotation is done using the "rotate" function. It takes an angle (in degrees) and an axis to rotate around. To make rotation easy, move an end to 0,0,0 first.

Translate the arm part down, then rotate it around Y:

...

```
// arms
rotate(a=30, v=[0,1,0])
  translate([0, 0, -4]) {
    cylinder(h = 4, r =
1);
    sphere(r=1);
  }
```

Note how we wrapped the arm+hand (cylinder+sphere) in curly braces to apply translate and rotate to both pieces.




10b) We'll need two of them, we can just mirror this part across the other side using "mirror":

...

```
// arms (1)
mirror([1,0,0])
  rotate(a=30, v=[0,1,0])
    translate([0, 0, -4]) {
      cylinder(h = 4, r = 1);
      sphere(r=1);
    }

// arms (2)
rotate(a=30, v=[0,1,0])
  translate([0, 0, -4]) {
    cylinder(h = 4, r = 1);
    sphere(r=1);
  }
```



11) Attach the arms to the body by moving them back up high enough to align with the cone-body, don't forget to wrap both arms in curly braces again:

```
$fn = 50;
// body
cylinder(h = 10, r1 = 4, r2 = 0);

// head
translate([0, 0, 9])
  sphere(r=2.5);

// arms
translate([0,0,7]) {
  mirror([1,0,0])
    rotate(a=35, v=[0,1,0])
      translate([0, 0, -4]) {
        cylinder(h = 4, r = 1);
        sphere(r=1);
      }

    rotate(a=35, v=[0,1,0])
      translate([0, 0, -4]) {
        cylinder(h = 4, r = 1);
        sphere(r=1);
      }
}
```

Done! You can modify it or design your own if you want. Make sure you keep the base flat so that it can be printed.

